

PHƯƠNG PHÁP CHIA ĐỂ TRỊ


1. Khái niệm

Chia để trị là phương pháp phân rã bài toán ban đầu thành các bài toán con “đồng dạng” và dễ giải quyết hơn. Các bài toán con tiếp tục được phân rã cho đến khi đủ nhỏ để có thể tìm được nghiệm trực tiếp và nghiệm của các bài toán con sẽ được tổng hợp để tìm nghiệm của bài toán ban đầu.

Bài toán con “đồng dạng” là bài toán ban đầu nhưng với bộ tham số hoặc ít hơn hoặc tồn tại 1 tham số có giá trị nhỏ hơn (bài toán ban đầu nhưng với kích thước nhỏ hơn).

Chia để trị dùng kỹ thuật đệ quy quay lui để phân rã thành các bài toán con và tổng hợp nghiệm của chúng.

Mô hình tổng quát của phương pháp chia để trị



2. Các bước tiếp cận xây dựng thuật toán chia để trị

Thuật toán chia để trị được thiết kế dựa trên 3 bước cơ bản sau:

- Chia: phân rã bài toán ban đầu thành các bài toán con đồng dạng
- Trị: giải các bài toán con bằng cách tiếp tục phân rã chúng một cách đệ quy
- Gộp: kết hợp nghiệm của các bài toán con để tìm nghiệm của bài toán ban đầu

Điều kiện cần để áp dụng phương pháp chia để trị:

- Phân rã được bài toán ban đầu thành các bài toán con đồng dạng
- Định nghĩa được mô hình liên hệ giữa bài toán ban đầu và các bài toán con
- Xác định được nghiệm của bài toán con nhỏ nhất (trường hợp cơ sở)

Thuật toán chia để trị tổng quát

```
DAC (P)
Begin
  If (P đủ nhỏ) then
    exit(nghiệm của P)
```

```
Else
Begin
    Chia P thành các bài toán con con p1, p2, ..., pk
    Kết hợp (DAC(p1), DAC(p2), ..., DAC(pk))
End;
End;
```

3. Đánh giá phương pháp chia để trị

Gọi $T(n)$ là thời gian giải bài toán kích thước n ; a là số bài toán con; n/b là kích thước của từng bài toán con; $f(n)$ là thời gian phân rã, kết hợp các bài toán con. Ta có

$$T(n) = a \cdot T\left(\frac{n}{b}\right) + f(n)$$

4. Một số bài toán minh họa

Bài toán 1: Tìm kiếm nhị phân

Ý tưởng:

- Không gian tìm kiếm ban đầu từ l đến r
- So sánh x với $a[m]$ (m là vị trí giữa đoạn l, r)
 - o Nếu $x = a[m]$ thì tìm kiếm thành công
 - o Nếu $x < a[m]$ thì tìm kiếm từ l đến $m - 1$
 - o Nếu $x > a[m]$ thì tìm kiếm từ $m + 1$ đến r

Chương trình

```
Function BinSearch(x, l, r: longint): longint;
Var m: longint;
Begin
    If (l > r) then Exit(-1);
    m := (l + r) div 2;
    If (x = a[m]) then Exit(m);
    If (x < a[m]) then Exit(BinSearch(x, l, m-1));
    Exit(BinSearch(x, m+1, r));
End;
```

Đánh giá

- Gọi $T(n)$ là thời gian tìm kiếm trên dãy gồm n phần tử, ta có

$$T(n) = T\left(\frac{n}{2}\right) + 1 = T(0) + \log n$$

Bài toán 2: Tính lũy thừa

Cho 2 số nguyên dương a, n . Tính a^n

Thuật toán 1: lặp với độ phức tạp $O(n)$

Thuật toán 2: chia để trị độ phức tạp $O(n)$

```
Function Pow(a, n: longint): longint;  
Begin  
  If (n = 0) then Exit(1);  
  Exit(a*Pow(a, n-1));  
End;
```

Thuật toán 3: chia để trị độ phức tạp $O(\log n)$

Nhận xét

- Trường hợp cơ sở $a^0 = 1$
- Nếu n chẵn thì $a^n = a^m * a^m, m = n \text{ div } 2$
- Nếu n lẻ thì $a^n = a * a^m * a^m, m = n \text{ div } 2$

Chương trình

```
Function Pow(a, n: longint): longint;  
Var t: longint;  
Begin  
  If (n = 0) then Exit(1);  
  t := Pow(a, n div 2);  
  if (n mod 2 = 0) then Exit(t*t);  
  Exit(a*t*t);  
End;
```

Đánh giá

- Gọi $T(n)$ là thời gian tính a^n của, ta có

$$T(n) = T(n \text{ div } 2) + 1 = T(0) + \log n$$

Bài toán 3: Tháp Hà Nội với n đĩa

Xét bài toán Hà Nội với 2 đĩa, 3 đĩa và tổng quát cho trường hợp n đĩa

Mô hình chia để trị chuyển n đĩa từ cột x sang cột z (thông qua cột trung gian y)

- Xem n đĩa như trường hợp 2 đĩa với đĩa trên cùng gồm $n-1$ đĩa
- Chuyển $n-1$ đĩa trên cùng từ cột x sang cột y
- Chuyển 1 đĩa từ cột x sang cột z
- Chuyển $n-1$ đĩa từ cột y sang cột z

Chương trình

```
Procedure Move(n, x, z: longint);
Var y: longint;
Begin
  If (n = 1) then writeln(x, '->', z)
  Else
    Begin
      Move(n-1, x, 6-x-z);
      Move(1, x, z);
      Move(n-1, 6-x-z, z);
    End;
End;
```

Đánh giá

Gọi $T(n)$ là thời gian giải bài toán chuyển n đĩa. Ta có

$$T(n) = 2 * T(n-1) + 1 = 2^n T(0) + (1 + 2 + 2^2 + \dots + 2^{n-1}) = 2^n - 1$$

Bài toán 4: Tìm số Fibonacci thứ n

Dãy số Fibonacci được định nghĩa theo công thức

$$F_n = \begin{cases} 1, & n \leq 2 \\ F_{n-1} + F_{n-2}, & n > 2 \end{cases}$$

```
Function Fibo(n: longint): longint;
Begin
  If (n <= 2) then Exit(1);
  Exit(Fibo(n-1) + Fibo(n-2));
End;
```

Nhận xét:

- $Fibo(n) = Fibo(n-1) + Fibo(n-2)$ và $Fibo(n-1) = Fibo(n-2) + Fibo(n-3)$. Trong lời gọi của $Fibo(n-1)$ sẽ lại gọi $Fibo(n-2)$ (đã được gọi trong $Fibo(n)$) do đó các bài toán con sẽ bị giải lại nhiều lần gây lãng phí thời gian và bộ nhớ.

Bài toán 5: Tính tổ hợp chập k của n phần tử

Tổ hợp chập k của n phần tử được tính theo công thức sau:

$$C_n^k = \begin{cases} 1, & n = k \text{ hoặc } k = 0 \\ C_{n-1}^k + C_{n-1}^{k-1} \end{cases}$$

```
Function C(n, k: longint): longint;
```

```
Begin
  If (k = 0) OR (n = k) then Exit(1);
  Exit(C(n-1,k) + C(n-1,k-1));
End;
```

Nhận xét:

- Lỗi gọi các bài toán con sẽ bị gọi lại nhiều lần gây tốn thời gian và bộ nhớ hệ thống

5. Kết luận

Chia để trị là một phương pháp tiếp cận để giải các bài toán có bản chất đệ quy. Kỹ thuật đệ quy giúp cho việc cài đặt thuật toán ngắn gọn và đơn giản. Tuy nhiên việc lạm dụng kỹ thuật đệ quy đôi khi làm cho chương trình trở nên chậm chạp tốn kém bộ nhớ.

Chia để trị không phù hợp đối với những bài toán mặc dù có bản chất đệ quy nhưng bài toán ban đầu bị phân rã thành các bài toán toán con có “giao nhau” (ví dụ bài tính tổ hợp, tìm số Fibonacci)

6. Bài tập

Cặp điểm láng giềng gần nhất

Trên mặt phẳng tọa độ Oxy cho n điểm phân biệt $(x_1, y_1), (x_2, y_2), \dots, (x_n, y_n)$. Hãy tìm 2 điểm có khoảng cách gần nhau nhất.